Interaktionseffekte: Was, wie, weshalb?

Kai Arzheimer

Übersicht

Stata code für heute: `https://www.kai-arzheimer.com/forschungsmethoden/interaktion.do`

Interaktionseffekte

Literatur

Brambor, T., Clark, W. R., & Golder, M. (2006). Understanding interaction models. improving empirical analyses. Political Analysis, 14(1), 63–82.

Linear-additive Modelle

- Normalerweise sind unsere Modelle „linear-additiv“ – Was heißt das?
- $y = \beta_0 + \beta_1 x_1 + \beta_2 x_2 + \cdots + \epsilon$

Linear-additive Modelle

- Normalerweise sind unsere Modelle „linear-additiv" – Was heißt das?
- $y = \beta_0 + \beta_1 x_1 + \beta_2 x_2 + \cdots + \epsilon$

Definition: „linear-additiv"

linear Effekt von x_1 ... ist für alle Werte von x_1 konstant

additiv Effekt von x_1 ... ist unabhängig vom Wert von x_2 ...

- Effekte aller Variablen werden einfach zusammengezählt
- Der Parameter für den Effekt von x_1 wird „unter Kontrolle" von x_2 ... geschätzt

Warum linear-additive Modelle?

- Soziale Realität: funktioniert weder linear noch additiv
- Aber Modelle sind nützliche *Vereinfachungen*
- Politikwissenschaftliche Theorien: schwach/unspezifisch
  - In +90% aller Fälle hinreichend gute Annäherung
  - Robust

Wie sieht das in 3D aus?



$$y = 2 + 1 \times x_1 + \frac{1}{2} \times x_2$$

Warum trotzdem Interaktionseffekte?

- ?
- *Manchmal* starke theoretische Gründe für Interaktion zweier Variablen
- Interpretation geht (immer noch) oft schief

Welche Typen von Interaktionen gibt es?

1. Effekte von x_1 und x_2 verstärken sich gegenseitig / schwächen sich gegenseitig ab
2. x_1 hat *nur dann* einen Effekt, wenn x_2 eine bestimmte Ausprägung hat (Moderation)
3. *Richtung* des Effektes von x_2 hängt davon ab, welchen Wert x_1 hat (Moderation)

Welche Typen von Interaktionen gibt es?

1. Effekte von x_1 und x_2 verstärken sich gegenseitig / schwächen sich gegenseitig ab
2. x_1 hat *nur dann* einen Effekt, wenn x_2 eine bestimmte Ausprägung hat (Moderation)
3. *Richtung* des Effektes von x_2 hängt davon ab, welchen Wert x_1 hat (Moderation)

- Alle drei Varianten sind mathematisch äquivalent
    - Man braucht immer Produktterme
    - Interaktive Modelle sind immer symmetrisch, ob man will oder nicht

Wie modelliert man Interaktionen?

- Für zwei Variablen $x_1, x_2 \rightarrow$ Produktterm = neue „künstliche" Variable, indem man beide miteinander multipliziert
- Auch möglich: Interaktion zwischen *drei* Variablen $\rightarrow$ insgesamt vier Produktterme
  - $x_1 \times x_2$
  - $x_1 \times x_3$
  - $x_2 \times x_3$
  - $x_1 \times x_2 \times x_3$
  - Nur in theoretisch sehr gut begründeten Fällen

Wichtig I

- Immer, immer, immer
- Alle Haupteffekte
- und Effekte für alle Produktvariablen schätzen
- (Gilt analog für kurvilineare Effekte etc. $(x_1 + x_1^2)$)

Wichtig I

- Immer, immer, immer
- Alle Haupteffekte
- und Effekte für alle Produktvariablen schätzen
- (Gilt analog für kurvilineare Effekte etc. $(x_1 + x_1^2)$)

Warum ist das wichtig?

- Variable im Modell nicht enthalten, die tatsächlich einen Effekt hat $\rightarrow$
- Schätzungen für Effekte aller Variablen verzerrt, die mit der ausgelassenen Variablen korreliert sind (*omitted variable bias*)
- *Produktterme sind per Definition mit Haupteffekten korreliert*

Warum ist das wichtig?

- Variable im Modell nicht enthalten, die tatsächlich einen Effekt hat $\rightarrow$
- Schätzungen für Effekte aller Variablen verzerrt, die mit der ausgelassenen Variablen korreliert sind (*omitted variable bias*)
- *Produktterme sind per Definition mit Haupteffekten korreliert*

- Haupteffekt für x_1 und Produktterm $x_1 \times x_2$ im Modell, aber nicht x_2
  - Nicht das, was Sie wahrscheinlich meinen (reine Moderation)
  - Sondern: Haupteffekt für $x_2 = 0$ wird erzwungen & Schätzungen für x_1 und $x_1 \times x_2$ unbrauchbar

Ein Beispiel

```
clear
set obs 5000
set seed 12345
gen epsilon = rnormal()*25
gen x1 = rnormal()  *3 +2
gen x2 = rnormal()*1+4
gen y = 5 + 2*x1 + 3*x2 + 4*x1*x2 + epsilon

* "Faktorvariablen" erleichtern den Umgang mit Interaktionen
* c.x1#c.x2 ist der Produktterm für die kontinuierlichen Variablen x1 und x2
* c.x1##c.x2 fügt automatisch die Hauptterme und den Produktterm ein
```

Ein Beispiel

Mit Interaktionseffekt

```
set cformat %9.3f
reg y c.x1##c.x2

      Source |       SS           df       MS      Number of obs   =     5,000
-------------+----------------------------------   F(3, 4996)      =   8888.93
       Model |  16350288.3         3  5450096.11   Prob > F        =    0.0000
    Residual |  3063212.97     4,996    613.1331   R-squared       =    0.8422
-------------+----------------------------------   Adj R-squared   =    0.8421
       Total |  19413501.3     4,999  3883.47695   Root MSE        =    24.762

------------------------------------------------------------------------------
           y |      Coef.   Std. Err.      t    P>|t|     [95% Conf. Interval]
-------------+----------------------------------------------------------------
          x1 |      2.187      0.484     4.52   0.000        1.238       3.136
          x2 |      2.710      0.420     6.45   0.000        1.886       3.534
             |
     c.x1#c.x2 |      3.975      0.117    34.12   0.000        3.747       4.203
             |
       _cons |      6.009      1.740     3.45   0.001        2.597       9.421
------------------------------------------------------------------------------
```

Ein Beispiel

Ohne Interaktionseffekt

```
set cformat %9.3f
reg y x1 x2

      Source |       SS           df       MS      Number of obs   =     5,000
-------------+----------------------------------   F(2, 4997)      =  10343.83
       Model |  15636567.3         2  7818283.66   Prob > F        =    0.0000
    Residual |  3776933.97     4,997  755.840298   R-squared       =    0.8054
-------------+----------------------------------   Adj R-squared   =    0.8054
       Total |  19413501.3     4,999  3883.47695   Root MSE        =    27.493

------------------------------------------------------------------------------
           y |      Coef.   Std. Err.      t    P>|t|     [95% Conf. Interval]
-------------+----------------------------------------------------------------
          x1 |     18.219      0.129   141.03   0.000       17.966      18.473
          x2 |     10.666      0.388    27.47   0.000        9.905      11.427
       _cons |    -26.023      1.627   -16.00   0.000      -29.212     -22.833
------------------------------------------------------------------------------
```

Ein Beispiel

Ohne zweiten Haupteffekt

```
set cformat %9.3f
reg y c.x1 c.x1#c.x2

      Source |       SS           df       MS      Number of obs   =     5,000
-------------+----------------------------------   F(2, 4997)      =  13205.42
       Model |  16324806.1         2  8162403.07   Prob > F        =    0.0000
    Residual |  3088695.15     4,997  618.109896   R-squared       =    0.8409
-------------+----------------------------------   Adj R-squared   =    0.8408
       Total |  19413501.3     4,999  3883.47695   Root MSE        =    24.862

------------------------------------------------------------------------------
           y |      Coef.   Std. Err.      t    P>|t|     [95% Conf. Interval]
-------------+----------------------------------------------------------------
          x1 |      0.510      0.410     1.24   0.214       -0.294       1.313
             |
     c.x1#c.x2 |      4.392      0.097    45.13   0.000        4.201       4.582
             |
       _cons |     16.888      0.427    39.57   0.000       16.051      17.725
------------------------------------------------------------------------------
```

Was bedeuten die Haupteffekte (nicht)?

- In der normalen linearen Regression: *nicht-konditionale* Effekte
    - Unabhängig vom Niveau dieser und aller anderen Variablen
    - Weil linear-additiv
- Hier: *konditionale* Effekte
    - Abhängig vom Niveau der anderen Ausgangsvariablen
    - Haupteffekt von x_1 = Effekt von $x_1 | x_2 = 0$
    - Oft nicht wahnsinnig interessant

Warum sind die Signifikanzen irrelevant?

- Signifikanz des Haupteffektes für x_1 gilt nur für $x_2 = 0$
- Signifikanz des gesamten Effektes = Signifikanz der Summe von Haupteffekt und Effekt des Produktterms *für gegebenes Niveau der anderen Variablen*
- Hängt ab von Standardfehler beider Terme und Kovarianz der Parameterschätzungen
- (Zentrierung wg Kollinearität unnötig, Schätzungen und deren Präzision sind identisch)
- Margins-Plots

Was bedeutet „marginaler Effekt“?

- Erwartete Veränderung von y, wenn sich x um eine Einheit verändert
- Im linear additiven Modell identisch mit Koeffizienten
- Im interaktiven oder nicht-linearen Modell abhängig vom Wert anderer/aller x-Variablen
    - x-Variablen auf mittlere/repräsentative Werte setzen
    - Entspricht Differenz vorhergesagten y-Werten (predictive margins)

3D-Beispiel von vorhin



$$y = 2 + 1 \times x_1 + \frac{1}{2} \times x_2$$

Mit den Daten aus unserem Beispiel von eben



$$y = 5 + 2 \times x_1 + 3 \times x_2 + 4 \times x_1 \times x_2$$

Margins + plots in Stata

```
reg y c.x1##c.x2

      Source |       SS           df       MS      Number of obs   =     5,000
-------------+----------------------------------   F(3, 4996)      =   8888.93
       Model |  16350288.3         3  5450096.11   Prob > F        =    0.0000
    Residual |  3063212.97     4,996    613.1331   R-squared       =    0.8422
-------------+----------------------------------   Adj R-squared   =    0.8421
       Total |  19413501.3     4,999  3883.47695   Root MSE        =    24.762

------------------------------------------------------------------------------
           y |      Coef.   Std. Err.      t    P>|t|     [95% Conf. Interval]
-------------+----------------------------------------------------------------
          x1 |      2.187      0.484     4.52   0.000        1.238       3.136
          x2 |      2.710      0.420     6.45   0.000        1.886       3.534
             |
     c.x1#c.x2 |    3.975      0.117    34.12   0.000        3.747       4.203
             |
       _cons |      6.009      1.740     3.45   0.001        2.597       9.421
------------------------------------------------------------------------------
```

Margins + plots in Stata

```
margins , dydx(x1) at(x2=(-3 -2 -1 0 1 2 3))

Average marginal effects                        Number of obs     =      5,000
Model VCE    : OLS

Expression   : Linear prediction, predict()
dy/dx w.r.t. : x1

1._at        : x2              =          -3

2._at        : x2              =          -2

3._at        : x2              =          -1

4._at        : x2              =           0

5._at        : x2              =           1

6._at        : x2              =           2

7._at        : x2              =           3

------------------------------------------------------------------------------
             |            Delta-method
             |      dy/dx   Std. Err.      t    P>|t|     [95% Conf. Interval]
-------------+----------------------------------------------------------------
x1           |
         _at |
          1  |     -9.738      0.828   -11.77   0.000      -11.360      -8.115
          2  |     -5.763      0.712    -8.09   0.000       -7.160      -4.366
          3  |     -1.788      0.598    -2.99   0.003       -2.960      -0.616
          4  |      2.187      0.484     4.52   0.000        1.238       3.136
          5  |      6.162      0.372    16.56   0.000        5.433       6.891
          6  |     10.137      0.264    38.41   0.000        9.620      10.654
          7  |     14.112      0.167    84.29   0.000       13.784      14.440
------------------------------------------------------------------------------
marginsplot

  Variables that uniquely identify margins: x2
```

Margins + plots in Stata



Margins + plots in Stata

```
margins , dydx(x2) at(x1=(-3 -2 -1 0 1 2 3))

Average marginal effects                        Number of obs     =      5,000
Model VCE    : OLS

Expression   : Linear prediction, predict()
dy/dx w.r.t. : x2

1._at        : x1              =          -3

2._at        : x1              =          -2

3._at        : x1              =          -1

4._at        : x1              =           0

5._at        : x1              =           1

6._at        : x1              =           2

7._at        : x1              =           3

------------------------------------------------------------------------------
             |            Delta-method
             |      dy/dx   Std. Err.      t    P>|t|     [95% Conf. Interval]
-------------+----------------------------------------------------------------
x2           |
         _at |
          1  |     -9.215      0.680   -13.56   0.000      -10.548      -7.883
          2  |     -5.240      0.583    -8.99   0.000       -6.383      -4.098
          3  |     -1.265      0.495    -2.56   0.011       -2.235      -0.296
          4  |      2.710      0.420     6.45   0.000        1.886       3.534
          5  |      6.685      0.369    18.13   0.000        5.962       7.407
          6  |     10.660      0.350    30.48   0.000        9.974      11.345
          7  |     14.635      0.369    39.71   0.000       13.912      15.357
------------------------------------------------------------------------------
marginsplot

  Variables that uniquely identify margins: x1
```

Margins + plots in Stata



Sie wollen statt dessen den y-Wert (die Steigung) plotten?

- Z.B. Steigung $y \leftrightarrow x_2$ für verschiedene Werte von x_1
- Wichtig: sinnvolle Werte wählen

```
summ x1 x2, det
margins , at(x2=(0/6) x1=(0 2 4 6) )
marginsplot
```

Sie wollen statt dessen den *y*-Wert (die Steigung) plotten?



Interaktionen und logistische Modelle

Wie funktionieren logistische Modelle?

- Einfachster Fall: binär-logistisch
  - Abhängige Variable: 0/1
  - Flashback: Maximum Likelihood-Schätzung (linearer Prädiktor + Funktion)
- *Wahrscheinlichkeit* für y=1: $\text{invlogit}(\beta_0 + \beta_1 x_1 + \beta_2 x_2 \dots)$
- Mit $\text{invlogit}(x) = \exp(x)/(1 + \exp(x))$
- Auf der Ebene der Logits sind die Effekte von $x_1, x_2, \dots$ linear-additiv
- Auf der Ebene der Wahrscheinlichkeiten hängt der Effekt von x_1 ab ...
  - Vom Wert von x_1 und ggf.
  - Vom Wert von $x_2, x_3, \dots \rightarrow$ margins
- D.h. eine Form der Interaktion ist „eingebaut“

Ein Beispiel

```
gen p = invlogit(-20 + 2*x1 + 3*x2)
set seed 12345
replace y = 0
replace y =1 if p >= runiform()
```

Logistische Regression

```
logit y x1 x2, nolog

Logistic regression                             Number of obs     =      5,000
                                                LR chi2(2)        =    4454.59
                                                Prob > chi2       =     0.0000
Log likelihood = -800.30144                     Pseudo R2         =     0.7357

------------------------------------------------------------------------------
           y |      Coef.   Std. Err.      z    P>|z|     [95% Conf. Interval]
-------------+----------------------------------------------------------------
          x1 |      2.033      0.077    26.50   0.000        1.883       2.184
          x2 |      3.000      0.125    23.97   0.000        2.755       3.245
       _cons |    -20.042      0.770   -26.03   0.000      -21.551     -18.533
------------------------------------------------------------------------------
Note: 88 failures and 2 successes completely determined.
```

Logistische Regression

```
margins ,dydx(x1)

Average marginal effects                        Number of obs     =      5,000
Model VCE    : OIM

Expression   : Pr(y), predict()
dy/dx w.r.t. : x1

------------------------------------------------------------------------------
             |            Delta-method
             |      dy/dx   Std. Err.      z    P>|z|     [95% Conf. Interval]
-------------+----------------------------------------------------------------
          x1 |      0.100      0.001   125.42   0.000        0.099       0.102
------------------------------------------------------------------------------
```

Logistische Regression

```
margins ,dydx(x1) at(x2=(3 5))

Average marginal effects                          Number of obs     =      5,000
Model VCE     : OIM

Expression    : Pr(y), predict()
dy/dx w.r.t. : x1

1._at         : x2              =           3

2._at         : x2              =           5

------------------------------------------------------------------------------
             |            Delta-method
             |      dy/dx   Std. Err.       z     P>|z|     [95% Conf. Interval]
-------------+----------------------------------------------------------------
x1           |
         _at |
          1  |      0.068      0.001    47.32   0.000         0.065       0.071
          2  |      0.127      0.000   346.83   0.000         0.126       0.128
------------------------------------------------------------------------------
```

Logistische Regression

```
margins ,dydx(x1) at(x1=(0 4))

Average marginal effects                        Number of obs     =      5,000
Model VCE     : OIM

Expression    : Pr(y), predict()
dy/dx w.r.t. : x1

1._at         : x1              =           0

2._at         : x1              =           4

------------------------------------------------------------------------------
             |            Delta-method
             |      dy/dx   Std. Err.      z    P>|z|     [95% Conf. Interval]
-------------+----------------------------------------------------------------
x1           |
         _at |
          1  |      0.017      0.001    13.05   0.000        0.014       0.019
          2  |      0.232      0.005    47.43   0.000        0.222       0.242
------------------------------------------------------------------------------
```

Logistische Regression

```
margins ,dydx(x1) predict(xb) at(x1=(1 2) x2=(3 5))

Conditional marginal effects                    Number of obs     =      5,000
Model VCE    : OIM

Expression   : Linear prediction (log odds), predict(xb)
dy/dx w.r.t. : x1

1._at        : x1              =           1
               x2              =           3

2._at        : x1              =           1
               x2              =           5

3._at        : x1              =           2
               x2              =           3

4._at        : x1              =           2
               x2              =           5

------------------------------------------------------------------------------
             |            Delta-method
             |      dy/dx   Std. Err.      z    P>|z|     [95% Conf. Interval]
-------------+----------------------------------------------------------------
x1           |
         _at |
          1  |      2.033      0.077    26.50   0.000        1.883       2.184
          2  |      2.033      0.077    26.50   0.000        1.883       2.184
          3  |      2.033      0.077    26.50   0.000        1.883       2.184
          4  |      2.033      0.077    26.50   0.000        1.883       2.184
------------------------------------------------------------------------------
```

Warum soll es überhaupt eine Interaktion *auf der Ebene der Logits* geben?

- Nur modellieren wenn aufgrund theoretischer Überlegungen geboten
- Beispiel: Zustimmung zu einer ethisch kontroversen Frage (z.B. Legalisierung von Euthanasie)
  - Nimmt mit säkularen Einstellungen zu
  - Aber nur/verstärkt, wenn Informationen vorhanden sind
  - Informationen selbst haben keinen/fast keinen Effekt (pure Moderation)
- In diesem Fall einfach Produktvariable bilden

Simulierte Daten

```
clear
set obs 2000
gen saek = 3*rnormal() +1
replace saek = -2 if saek <-2
replace saek = 2 if saek >2
gen byte inform = runiform() >.6
gen p = invlogit(-0.1 + .7*saek*inform)
gen euth = 0
replace euth =1 if p >= runiform()
```

Regressionsmodell

```
logit euth c.saek##i.inf, nolog

Logistic regression                             Number of obs     =      2,000
                                                LR chi2(3)        =     210.30
                                                Prob > chi2       =     0.0000
Log likelihood = -1281.0615                     Pseudo R2         =     0.0759

-------------------------------------------------------------------------------
         euth |      Coef.   Std. Err.      z    P>|z|     [95% Conf. Interval]
--------------+----------------------------------------------------------------
         saek |      0.007      0.037     0.19   0.847       -0.066       0.080
     1.inform |     -0.057      0.106    -0.53   0.593       -0.264       0.151
              |
inform#c.saek |
           1  |      0.681      0.065    10.40   0.000        0.552       0.809
              |
        _cons |     -0.094      0.061    -1.53   0.125       -0.214       0.026
-------------------------------------------------------------------------------
```

Fazit

Fazit

- Linear-additive Modelle: besonders unrealistisch, aber robust und nützlich
- Theoretisch-inhaltliche Gründe $\rightarrow$ interaktives Modell
- Modellierung selbst über Produktvariable ist trivial
- Wichtig
  - Immer Produktterm und Haupteffekte schätzen
  - Signifikanz/Standardfehler von Haupteffekten und Interaktionsterm ignorieren
  - Statt dessen auf Verlauf/Signifikanz der *marginalen Effekte* konzentrieren